

Introduction to VHDL

Wenye Li, Ph.D

The Chinese University of Hong Kong, Shenzhen

Note

- A short discussion of VHDL within <3 hours.
 - A full investigation needs one semester or more.
- Two ways learning a programming language:
 - Learn through grammars (slow)
 - Learn through examples/practices (quick)
 - This course, firstly learn by examples, then grammars
- Some materials are NOT directly from the textbook!

Outline

- **VHDL Examples**
- VHDL Language
- Advanced Examples





FIGURE 3-55 Programming setup for reprogrammable logic devices. (Photo courtesy of Digilent, Inc.)

Text Entry and Graphic Entry

```
Vhdl1.vhd
entity VHDL1 is
  port(A, B: in bit; X: out bit);
end entity VHDL1;
architecture ANDfunction of VHDL1 is
begin
  X <= A and B;
end architecture ANDfunction;
```


(a) VHDL text entry




(b) Equivalent graphic (schematic) entry

FIGURE 3-56 Examples of design entry of an AND gate.


Basic Gates




entity Inverter **is**
port (A: **in** bit; X: **out** bit);
end entity Inverter;
architecture NOTfunction **of** Inverter **is**
begin
 X <= not A;
end architecture NOTfunction;
 (a) Inverter




entity ANDgate **is**
port (A, B: **in** bit; X: **out** bit);
end entity ANDgate;
architecture ANDfunction **of** ANDgate **is**
begin
 X <= A and B;
end architecture ANDfunction;
 (b) AND gate



entity ORgate **is**
port (A, B: **in** bit; X: **out** bit);
end entity ORgate;
architecture ORfunction **of** ORgate **is**
begin
 X <= A or B;
end architecture ORfunction;
 (c) OR gate




entity NANDgate **is**
port (A, B, C: **in** bit; X: **out** bit);
end entity NANDgate;
architecture NANDfunction **of** NANDgate **is**
begin
 X <= A nand B nand C;
end architecture NANDfunction;
 (d) NAND gate



entity XNORgate **is**
port (A, B: **in** bit; X: **out** bit);
end entity XNORgate;
architecture XNORfunction **of** XNORgate **is**
begin
 X <= A xnor B;
end architecture XNORfunction;
 (e) XNOR gate


FIGURE 3-58 Logic gates described with VHDL.

Entity



```
ENTITY Full_adder IS
  PORT (
    -- I/O ports
    a:   IN STD_LOGIC;    -- a input
    b:   IN STD_LOGIC;    -- b input
    cin: IN STD_LOGIC;    -- carry input
    sum: OUT STD_LOGIC;   -- sum output
    cout: OUT STD_LOGIC); -- carry output
END Full_adder ;
```

Architecture



ARCHITECTURE dataflow OF Full_adder IS


BEGIN

sum <= a xor b xor cin;

cout <= (a and b) or (a and cin) or
(b and cin);

END dataflow;

Component



entity AND_OR_Logic **is**

port (IN1, IN2, IN3, IN4: **in** bit; OUT3: **out** bit);

end entity AND_OR_Logic;

architecture LogicOperation **of** AND_OR_Logic **is**

```
component AND_gate is  
  port (A, B: in bit; X: out bit);  
end component AND_gate;
```

← Component declaration for the
AND gate

```
component OR_gate is  
  port (A, B: in bit; X: out bit);  
end component OR_gate;
```

← Component declaration for the
OR gate

```
signal OUT1, OUT2: bit;
```

← Signal declaration

begin

```
G1: AND_gate port map (A => IN1, B => IN2, X => OUT1);
```


```
G2: AND_gate port map (A => IN3, B => IN4, X => OUT2);
```

```
G3: OR_gate port map (A => OUT1, B => OUT2, X => OUT3);
```

← Component instantiations describe
how the three gates are connected.

end architecture LogicOperation;

Component Initialization



G1: AND_gate **port map** (A => IN1, B => IN2, X => OUT1);

G2: AND_gate **port map** (A => IN3, B => IN4, X => OUT2);

G3: OR_gate **port map** (A => OUT1, B => OUT2, X => OUT3);

Component instantiations describe
how the three gates are connected.

Exercise

Write a VHDL program for the SOP logic circuit in Figure 5–41 using the structural approach and compare with the data flow approach. Assume that VHDL components for a 3-input NAND gate and for a 2-input NAND are available. Notice the NAND gate G4 is shown as a negative-OR.




FIGURE 5-41

Structural Approach

entity SOP_Logic **is**

port (IN1, IN2, IN3, IN4, IN5, IN6, IN7, IN8: **in** bit; OUT4: **out** bit);

end entity SOP_Logic;

architecture LogicOperation **of** SOP_Logic **is**

--component declaration for 3-input NAND gate

component NAND_gate3 **is**

port (A, B, C: **in** bit X: **out** bit);

end component NAND_gate3;

--component declaration for 2-input NAND gate

component NAND_gate2 **is**

port (A, B: **in** bit; X: **out** bit);

end component NAND_gate2;

signal OUT1, OUT2, OUT3: bit;

begin

G1: NAND_gate3 **port map** (A => IN1, B => IN2, C => IN3, X => OUT1);

G2: NAND_gate3 **port map** (A => IN4, B => IN5, C => IN6, X => OUT2);

G3: NAND_gate2 **port map** (A => IN7, B => IN8, X => OUT3);

G4: NAND_gate3 **port map** (A => OUT1, B => OUT2, C => OUT3, X => OUT4);

end architecture LogicOperation;


Data Flow Approach

```
entity SOP_Logic is
    port (IN1, IN2, IN3, IN4, IN5, IN6, IN7, IN8: in bit; OUT4: out bit);
end entity SOP_Logic;

architecture LogicOperation of SOP_Logic is
begin

    OUT4 <= (IN1 and IN2 and IN3) or (IN4 and IN5 and IN6) or (IN7 and IN8);
end architecture LogicOperation;
```

VHDL Development Environment



```
entity Combinational is
  port ( A, B, C, D: in bit; X, Y: out bit );
end entity Combinational;

architecture Example of Combinational is
begin
  X <= ( A and B ) or not C;
  Y <= C or not D;
end architecture Example;
```

FIGURE 5-42 A VHDL program for a combinational logic circuit after entry on a generic text editor screen that is part of a software development tool.




FIGURE 5-43 A typical waveform editor tool showing the simulated waveforms for the logic circuit described by the VHDL code in Figure 5-42.

Outline

- VHDL Examples
- **VHDL Language**
- Advanced Examples


Libraries and Packages

- Libraries provide a set of packages, components, and functions that simplify the task of designing hardware
- Packages provide a collection of related data types and subprograms
- The following is an example of the use of the `ieee` library and its `std_logic_1164` package:

```
LIBRARY ieee;  
  
USE ieee.std_logic_1164.ALL;
```


Entities, Architectures and Configurations

- The structure of a VHDL design resembles the structure of a modern, object-oriented software design
- All VHDL designs provide an external interface and an internal implementation
- A VHDL design consists of entities, architectures, and configurations



Entities

- An entity is a specification of the design's external interface
- Entity declarations specify the following:
 1. The name of the entity
 2. A set of generic declarations specifying instance-specific parameters
 3. A set of port declarations defining the inputs and outputs of the hardware design
- Generic declarations and port declarations are optional

Entities

- Entity declarations are specified as follows:

```
ENTITY entity_name IS
    GENERIC (
        generic_1_name      : generic_1_type;
        generic_2_name      : generic_2_type;
        generic_n_name      : generic_n_type
    );
    PORT (
        port_1_name         : port_1_dir  port_1_type;
        port_2_name         : port_2_dir  port_2_type;
        port_n_name         : port_n_dir  port_n_type
    );
END entity_name;
```

Ports

- Port name choices:
 - Consist of letters, digits, and/or underscores
 - Always begin with a letter
 - Case insensitive

- Port direction choices:

IN	Input port
OUT	Output port
INOUT	Bidirectional port
BUFFER	Buffered output port

NOTE:

A buffer is an output that can be “read” by the architecture of the entity.

Ports

- IEEE standard 1164-1993 defines a package which provides a set of data types that are useful for logic synthesis
 - The external pins of a synthesizable design must use data types specified in the `std_logic_1164` package
 - IEEE recommends the use of the following data types to represent signals in a synthesizable system:

`std_logic`

`std_logic_vector(<max> DOWNTO <min>)`

Architecture

- An architecture is a specification of the design's internal implementation
- Multiple architectures can be created for a particular entity
- For example, you might wish to create several architectures for a particular entity with each architecture optimized with respect to a design goal:
 - Performance
 - Area
 - Power Consumption
 - Ease of Simulation

Architecture

- Architecture declarations are specified as follows:

```
ARCHITECTURE architecture_name OF entity_name IS
BEGIN
    -- Insert VHDL statements to assign outputs to
    -- each of the output signals defined in the
    -- entity declaration.
END architecture_name;
```

Configurations

- A configuration is a specification of the mapping between an architecture and a particular instance of an entity
- By default, a configuration exists for each entity
- The default configuration maps the most recently compiled architecture to the entity
- Configurations are most often used to specify alternative architectures for hardware designs

Signals

- Signals represent wires and storage elements
- Signals may only be defined inside architectures
- Signals are associated with a data type
- Signals have attributes
- VHDL is a strongly-typed language:
 - Explicit type conversion is supported
 - Implicit type conversion is not supported

Signal Representations

- Binary number representations are sufficient for software programming languages

Binary			
Forcing 1	'1'	Forcing 0	'0'

- Physical wires cannot be modelled accurately using a binary number representation
- Additional values are necessary to accurately represent the state of a wire

Built-In Data Types

- VHDL supports a rich set of built-in data types as well as user-defined data types

Data Type	Characteristics
BIT	Binary, Unresolved
BIT_VECTOR	Binary, Unresolved, Array
INTEGER	Binary, Unresolved, Array
REAL	Floating Point

- Built-in data types work well for simulation but not so well for synthesis
- Built-in data types are suitable for use inside an architecture but should not be used for external pins

Logical Operators

- VHDL supports the following logical operators:

AND	NAND	NOT
OR	NOR	
XOR	XNOR	

- VHDL also supports the overloading of existing operators and the creation of new operators using functions

Other Operators

- VHDL supports the following relational operators:
 - = (Equal)
 - /= (Not Equal)
 - < (Less Than)
 - > (Greater Than)
- VHDL supports the following mathematical operators:
 - + (Addition)
 - (Subtraction)
 - * (Multiplication)
 - / (Division)

Assignment Statements

```
SIGNAL a, b, c          : std_logic;
SIGNAL avec, bvec, cvec  : std_logic_vector(7 DOWNT0 0);

-- Concurrent Signal Assignment Statements
-- NOTE: Both a and avec are produced concurrently
a      <= b AND c;
avec   <= bvec OR cvec;

-- Alternatively, signals may be assigned constants
a      <= '0';
b      <= '1';
c      <= 'Z';
avec   <= "00111010";      -- Assigns 0x3A to avec
bvec   <= X"3A";           -- Assigns 0x3A to bvec
cvec   <= X"3" & X"A";     -- Assigns 0x3A to cvec
```

Assignment Statements

```
SIGNAL a, b, c, d          :std_logic;
SIGNAL avec                :std_logic_vector(1 DOWNTO 0);
SIGNAL bvec                :std_logic_vector(2 DOWNTO 0);

-- Conditional Assignment Statement
-- NOTE: This implements a tree structure of logic gates
a <=      '0'      WHEN avec = "00" ELSE
        b        WHEN avec = "11" ELSE
        c        WHEN d = '1' ELSE
        '1';

-- Selected Signal Assignment Statement
-- NOTE: The selection values must be constants
bvec <= d & avec;
WITH bvec SELECT
a <=      '0'      WHEN "000",
        b        WHEN "011",
        c        WHEN "1--",      -- Some tools won't synthesize '-' properly
        '1'      WHEN OTHERS;
```

Assignment Statements

```
SIGNAL a                                :std_logic;
SIGNAL avec, bvec                       :std_logic_vector(7 DOWNTO 0);

-- Selected Signal Assignment Statement
-- NOTE: Selected signal assignments also work
--       with vectors
WITH a SELECT
avec <=  "01010101"                     WHEN '1',
        bvec                               WHEN OTHERS;
```


Process Statements

- VHDL supports processes
- Processes encapsulate a portion of a design
- Processes have a sensitivity list that specifies signals and ports that cause changes in the outputs of the process
 - Sensitivity lists can be used to preserve the state of a hardware system
- For example, an edge-triggered flip-flop circuit is sensitive to a particular clock edge
 - The output of the edge-triggered flip-flop changes if and only if a particular clock edge is received
 - Otherwise, the previous output remains asserted

Process Statements

- The keywords used for conditional assignments and selected assignments differ from those used within a process:

Outside Processes	Inside Processes
WHEN..ELSE	IF..ELSIF..ELSE..END IF
WITH..SELECT..WHEN	CASE..WHEN..END CASE

- A selected assignment outside a process is functionally equivalent to a case statement within a process
- Processes can be used for combinational logic but most often, processes encapsulate sequential logic

Process Statements

```
SIGNAL reset, clock, d, q           :std_logic;

PROCESS (reset, clock)
-- reset and clock are in the sensitivity list to
-- indicate that they are important inputs to the process
BEGIN
    -- IF keyword is only valid in a process
    IF (reset = '0') THEN
        q <= 0;
    ELSIF (clock'EVENT AND clock = '1') THEN
        q <= d;
    END IF;
END PROCESS;
```

The **EVENT** attribute is true if an edge has been detected on the corresponding signal.

NOTE:

This implements a D flip-flop with an asynchronous active-low reset signal.

Process Statements

```
SIGNAL a, b, c, d          :std_logic;


PROCESS (a, b, d)
-- a, b, and d are in the sensitivity list to indicate that
-- the outputs of the process are sensitive to changes in them
BEGIN
    -- CASE keyword is only valid in a process
    CASE d IS
        WHEN '0' =>
            c <= a AND b;
        WHEN OTHERS =>
            c <= '1';
    END CASE;
END PROCESS;
```

NOTE:

This implements a combinational circuit.

D Flip-Flop Example

- Using a process and the `EVENT` attribute of a signal, it is possible to specify a D flip-flop
- The `EVENT` attribute can be used to check for the rising edge of a clock signal
- The block diagram of a D Flip-Flop is shown below:



D Flip-Flop Example

```
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;

ENTITY dffe IS
    PORT(rst, clk, ena, d : IN    std_logic;
          q                : OUT  std_logic );
END dffe;

ARCHITECTURE synthesis1 OF dffe IS
BEGIN
    PROCESS (rst, clk)
    BEGIN
        IF (rst = '1') THEN
            q <= '0';
        ELSIF (clk'EVENT) AND (clk = '1') THEN
            IF (ena = '1') THEN
                q <= d;
            END IF;
        END IF;
    END PROCESS;
END synthesis1;
```

Outline

- VHDL Examples
- VHDL Language
- **Advanced Examples**

encoder_using_if.vhd

```
1 -----
2 -- Design Name : encoder_using_if
3 -- File Name   : encoder_using_if.vhd
4 -- Function    : Encoder using If
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity encoder_using_if is
11     port (
12         enable      :in std_logic;           -- Enable for the encoder
13         encoder_in   :in std_logic_vector (15 downto 0); -- 16-bit Input
14         binary_out  :out std_logic_vector (3 downto 0) -- 4 bit binary Output
15     );
16 end entity;
17
18 architecture behavior of encoder_using_if is
19 begin
20     process (enable, encoder_in) begin
21         binary_out <= "0000";
22         if (enable = '1') then
23             if (encoder_in = X"0002") then binary_out <= "0001"; end if;
24             if (encoder_in = X"0004") then binary_out <= "0010"; end if;
25             if (encoder_in = X"0008") then binary_out <= "0011"; end if;
26             if (encoder_in = X"0010") then binary_out <= "0100"; end if;
27             if (encoder_in = X"0020") then binary_out <= "0101"; end if;
28             if (encoder_in = X"0040") then binary_out <= "0110"; end if;
29             if (encoder_in = X"0080") then binary_out <= "0111"; end if;
30             if (encoder_in = X"0100") then binary_out <= "1000"; end if;
31             if (encoder_in = X"0200") then binary_out <= "1001"; end if;
32             if (encoder_in = X"0400") then binary_out <= "1010"; end if;
33             if (encoder_in = X"0800") then binary_out <= "1011"; end if;
34             if (encoder_in = X"1000") then binary_out <= "1100"; end if;
35             if (encoder_in = X"2000") then binary_out <= "1101"; end if;
36             if (encoder_in = X"4000") then binary_out <= "1110"; end if;
37             if (encoder_in = X"8000") then binary_out <= "1111"; end if;
38         end if;
39     end process;
40 end architecture;
```



```
1 -----
2 -- Design Name : encoder_using_case
3 -- File Name   : encoder_using_case.vhd
4 -- Function    : Encoder using Case
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity encoder_using_case is
11     port (
12         enable    :in std_logic;           -- Enable for the encoder
13         encoder_in :in std_logic_vector (15 downto 0); -- 16-bit Input
14         binary_out :out std_logic_vector (3 downto 0) -- 4 bit binary Output
15     );
16 end entity;
17
18 architecture behavior of encoder_using_case is
19 begin
20
21     process (enable, encoder_in) begin
22         if (enable = '1') then
23             case (encoder_in) is
24                 when X"0002" => binary_out <= "0001";
25                 when X"0004" => binary_out <= "0010";
26                 when X"0008" => binary_out <= "0011";
27                 when X"0010" => binary_out <= "0100";
28                 when X"0020" => binary_out <= "0101";
29                 when X"0040" => binary_out <= "0110";
30                 when X"0080" => binary_out <= "0111";
31                 when X"0100" => binary_out <= "1000";
32                 when X"0200" => binary_out <= "1001";
33                 when X"0400" => binary_out <= "1010";
34                 when X"0800" => binary_out <= "1011";
35                 when X"1000" => binary_out <= "1100";
36                 when X"2000" => binary_out <= "1101";
37                 when X"4000" => binary_out <= "1110";
38                 when X"8000" => binary_out <= "1111";
39                 when others => binary_out <= "0000";
40             end case;
41         end if;
42     end process;
43 end architecture;
```

```
< > encoder_using_if.vhd × encoder_using_case.vhd • mux_using_with.vhd •
1 -----
2 -- Design Name : mux_using_with
3 -- File Name   : mux_using_with.vhd
4 -- Function    : 2:1 Mux using with-select
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity mux_using_with is
11     port (
12         din_0  :in std_logic; -- Mux first input
13         din_1  :in std_logic; -- Mux Second input
14         sel    :in std_logic; -- Select input
15         mux_out :out std_logic -- Mux output
16     );
17 end entity;
18
19
20 architecture behavior of mux_using_with is
21
22 begin
23     with (sel) select
24         mux_out <= din_0 when '0',
25                 din_1 when others;
26
27 end architecture;
28
```

```
< > encoder_using_if.vhd × encoder_using_case.vhd × mux_using_with.vhd × dlatch_reset.vhd ×
1 -----
2 -- Design Name : dlatch_reset
3 -- File Name   : dlatch_reset.vhd
4 -- Function    : DLATCH async reset
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity dlatch_reset is
11     port (
12         data :in std_logic; -- Data input
13         en   :in std_logic; -- Enable input
14         reset :in std_logic; -- Reset input
15         q    :out std_logic -- Q output
16     );
17 end entity;
18
19 architecture rtl of dlatch_reset is
20 begin
21     process (en, reset, data) begin
22         if (reset = '0') then
23             q <= '0';
24         elsif (en = '1') then
25             q <= data;
26         end if;
27     end process;
28 end architecture;
```

```
< > encoder_using_if.vhd × encoder_using_case.vhd × mux_using_with.vhd × dlatch_reset.vhd × dff_async_reset.vhd ×
1 -----
2 -- Design Name : dff_async_reset
3 -- File Name   : dff_async_reset.vhd
4 -- Function    : D flip-flop async reset
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 |-----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity dff_async_reset is
11     port (
12         data  :in std_logic; -- Data input
13         clk   :in std_logic; -- Clock input
14         reset :in std_logic; -- Reset input
15         q     :out std_logic -- Q output
16     );
17 end entity;
18
19
20 architecture rtl of dff_async_reset is
21
22 begin
23     process (clk, reset) begin
24         if (reset = '0') then
25             q <= '0';
26         elsif (rising_edge(clk)) then
27             q <= data;
28         end if;
29     end process;
30
31 end architecture;
```

```
1 -----
2 -- Design Name : dff_sync_reset
3 -- File Name   : dff_sync_reset.vhd
4 -- Function    : D flip-flop sync reset
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity dff_sync_reset is
11     port (
12         data :in std_logic; -- Data input
13         clk   :in std_logic; -- Clock input
14         reset :in std_logic; -- Reset input
15         q     :out std_logic -- Q output
16     );
17 end entity;
18
19 architecture rtl of dff_sync_reset is
20
21 begin
22     process (clk) begin
23         if (rising_edge(clk)) then
24             if (reset = '0') then
25                 q <= '0';
26             else
27                 q <= data;
28             end if;
29         end if;
30     end process;
31 end architecture;
```

```
1 -----
2 -- Design Name : up_counter
3 -- File Name   : up_counter.vhd
4 -- Function    : Up counter
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9     use ieee.std_logic_unsigned.all;
10
11 entity up_counter is
12     port (
13         cout    :out std_logic_vector (7 downto 0); -- Output of the counter
14         enable  :in  std_logic;                -- Enable counting
15         clk     :in  std_logic;                -- Input clock
16         reset   :in  std_logic                -- Input reset
17     );
18 end entity;
19
20 architecture rtl of up_counter is
21     signal count :std_logic_vector (7 downto 0);
22 begin
23     process (clk, reset) begin
24         if (reset = '1') then
25             count <= (others=>'0');
26         elsif (rising_edge(clk)) then
27             if (enable = '1') then
28                 count <= count + 1;
29             end if;
30         end if;
31     end process;
32     cout <= count;
33 end architecture;
```

```
1 -----
2 -- Design Name : gray_counter
3 -- File Name   : gray_counter.vhd
4 -- Function    : 8 bit gray counters
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9     use ieee.std_logic_unsigned.all;
10
11 entity gray_counter is
12     port (
13         cout    :out std_logic_vector (7 downto 0); -- Output of the counter
14         enable  :in  std_logic;                -- Enable counting
15         clk     :in  std_logic;                -- Input clock
16         reset   :in  std_logic;                -- Input reset
17     );
18 end entity;
19
20 architecture rtl of gray_counter is
21     signal count :std_logic_vector (7 downto 0);
22 begin
23     process (clk, reset) begin
24         if (reset = '1') then
25             count <= (others=>'0');
26         elsif (rising_edge(clk)) then
27             if (enable = '1') then
28                 count <= count + 1;
29             end if;
30         end if;
31     end process;
32     cout <= (count(7) &
33             (count(7) xor count(6)) &
34             (count(6) xor count(5)) &
35             (count(5) xor count(4)) &
36             (count(4) xor count(3)) &
37             (count(3) xor count(2)) &
38             (count(2) xor count(1)) &
39             (count(1) xor count(0)) );
40 end architecture;
41
```

```
1 -----
2 -- Design Name : parity_using_assign
3 -- File Name   : parity_using_assign.vhd
4 -- Function    : Parity using direct assignment
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity parity_using_assign is
11     port (
12         data_in      :in std_logic_vector (7 downto 0);
13         parity_out    :out std_logic
14     );
15 end entity;
16
17 architecture rtl of parity_using_assign is
18
19 begin
20
21     parity_out <= (data_in(0) xor data_in(1)) xor
22                 (data_in(2) xor data_in(3)) xor
23                 (data_in(4) xor data_in(5)) xor
24                 (data_in(6) xor data_in(7));
25 end architecture;
```



```
1 -----
2 -- Design Name : parity_using_function
3 -- File Name   : parity_using_function.vhd
4 -- Function    : Parity using function
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity parity_using_function is
11     port (
12         data_in      :in std_logic_vector (7 downto 0);
13         parity_out    :out std_logic
14     );
15 end entity;
16
17 architecture rtl of parity_using_function is
18
19     function parity(d_in :std_logic_vector) return std_logic is
20         variable result :std_logic;
21     begin
22         result := (d_in(0) xor d_in(1)) xor
23                 (d_in(2) xor d_in(3)) xor
24                 (d_in(4) xor d_in(5)) xor
25                 (d_in(6) xor d_in(7));
26
27         return result;
28
29     end parity;
30
31
32 begin
33     -- Function Call
34     parity_out <= parity(data_in);
35
36 end architecture;
```

```
1 -----
2 -- Design Name : parity_using_function2
3 -- File Name   : parity_using_function2.vhd
4 -- Function    : Parity using function
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity parity_using_function2 is
11     port (
12         data_in      :in  std_logic_vector (31 downto 0);
13         parity_out    :out std_logic
14     );
15 end entity;
16
17 architecture rtl of parity_using_function2 is
18
19     function parity (d_in :std_logic_vector) return std_logic is
20         variable result :std_logic;
21     begin
22         result := '0';
23         for i in 0 to d_in'length-1 loop
24             result := result xor d_in(i);
25         end loop;
26
27         return result;
28
29     end parity;
30
31
32 begin
33     -- Function Call
34     parity_out <= parity(data_in);
35
36 end architecture;
37
```

```
1 -----
2 -- Design Name : parity_using_bitwise
3 -- File Name   : parity_using_bitwise.vhd
4 -- Function    : Parity using bitwise xor
5 -- Prepared by : Wenye Li for EIE2050 at CUHK-SZ
6 -----
7 library ieee;
8     use ieee.std_logic_1164.all;
9
10 entity parity_using_bitwise is
11     port (
12         data_in      :in  std_logic_vector (7 downto 0);
13         parity_out    :out std_logic
14     );
15 end entity;
16
17 architecture rtl of parity_using_bitwise is
18
19 begin
20
21     process (data_in)
22         variable pbit :std_logic;
23     begin
24         pbit := '0';
25         for i in 0 to 7 loop
26             pbit := pbit xor data_in(i);
27         end loop;
28
29         parity_out <= pbit;
30
31     end process;
32
33 end architecture;
```

Thanks